

Journal

Clinical Pharmacokinetics

Title

Neonatal Maturation of Paracetamol (Acetaminophen) Glucuronidation, Sulfation, and Oxidation Based on a Parent–Metabolite Population Pharmacokinetic Model

Authors and affiliations

Sarah F. Cook¹, Chris Stockmann^{2,3}, Samira Samiee-Zafarghandy^{4,5}, Amber D. King¹, Nina Deutsch⁶, Elaine F. Williams⁴, Diana G. Wilkins^{1,7}, Catherine M. T. Sherwin^{2,3,8}, John N. van den Anker^{4,9,10,11}

¹ Center for Human Toxicology, Department of Pharmacology and Toxicology, University of Utah, Salt Lake City, UT, USA

² Division of Clinical Pharmacology, Department of Pediatrics, University of Utah School of Medicine, Salt Lake City, UT, USA

³ Department of Pharmacology and Toxicology, University of Utah, Salt Lake City, UT, USA

⁴ Division of Clinical Pharmacology, Children’s National Health System, Washington, DC, USA

⁵ Division of Neonatology, Department of Pediatrics, McMaster University, Hamilton, ON, Canada

⁶ Division of Anesthesiology, Sedation, and Perioperative Medicine, Children's National Health System, Washington, DC, USA

⁷ Division of Medical Laboratory Sciences, Department of Pathology, University of Utah School of Medicine, Salt Lake City, UT, USA

⁸ Clinical Trials Office, Department of Pediatrics, University of Utah School of Medicine, Salt Lake City, UT, USA

⁹ Departments of Pediatrics, Integrative Systems Biology, Pharmacology and Physiology, George Washington University School of Medicine and Health Sciences, Washington, DC, USA

¹⁰ Intensive Care and Department of Pediatric Surgery, Erasmus Medical Center–Sophia Children’s Hospital, Rotterdam, The Netherlands

¹¹ Department of Paediatric Pharmacology, University Children’s Hospital Basel, Basel, Switzerland

Corresponding author

Catherine M. T. Sherwin

E-mail: catherine.sherwin@hsc.utah.edu




Fig. S1 Dosing and pharmacokinetic sampling schemes. Neonates <28 weeks' gestation received five doses at 12-h intervals; neonates ≥ 28 weeks' gestation received seven doses at 8-h intervals. Patients were randomly assigned to one of two plasma sampling schedules, each consisting of 9–10 collection times.




Fig. S2 Plots of plasma paracetamol NPDEs for evaluation of the final model. Density histogram of NPDEs (a) with overlaid *solid black curve* depicting the standard normal distribution. NPDEs versus **b** time since first dose, **c** population-predicted concentration, **d** weight, and **e** postnatal age. In **b–e**, the *dashed red lines* depict the 5th, 50th, and 95th percentiles of the NPDEs and the *solid black lines* depict the 5th, 50th, and 95th percentiles of the standard normal distribution. *NPDEs* normalized prediction distribution errors




Fig. S3 Plots of plasma paracetamol-glucuronide NPDEs for evaluation of the final model. Density histogram of NPDEs (a) with overlaid *solid black curve* depicting the standard normal distribution. NPDEs versus **b** time since first dose, **c** population-predicted concentration, **d** weight, and **e** postnatal age. In **b–e**, the *dashed red lines* depict the 5th, 50th, and 95th percentiles of the NPDEs and the *solid black lines* depict the 5th, 50th, and 95th percentiles of the standard normal distribution. *NPDEs* normalized prediction distribution errors




Fig. S4 Plots of plasma paracetamol-sulfate NPDEs for evaluation of the final model. Density histogram of NPDEs (a) with overlaid *solid black curve* depicting the standard normal distribution. NPDEs versus **b** time since first dose, **c** population-predicted concentration, **d** weight, and **e** postnatal age. In **b–e**, the *dashed red lines* depict the 5th, 50th, and 95th percentiles of the NPDEs and the *solid black lines* depict the 5th, 50th, and 95th percentiles of the standard normal distribution. *NPDEs* normalized prediction distribution errors




Fig. S5 Plots of plasma oxidative pathway metabolite NPDEs for evaluation of the final model. Density histogram of NPDEs (a) with overlaid *solid black curve* depicting the standard normal distribution. NPDEs versus **b** time since first dose, **c** population-predicted concentration, **d** weight, and **e** postnatal age. In **b–e**, the *dashed red lines* depict the 5th, 50th, and 95th percentiles of the NPDEs and the *solid black lines* depict the 5th, 50th, and 95th percentiles of the standard normal distribution. *NPDEs* normalized prediction distribution errors




Fig. S6 Plots of urinary paracetamol NPDEs for evaluation of the final model. Density histogram of NPDEs (a) with overlaid *solid black curve* depicting the standard normal distribution. NPDEs versus **b** time since first dose, **c** population-predicted amount, **d** weight, and **e** urine flow rate. In **b–e**, the *dashed red lines* depict the 5th, 50th, and 95th percentiles of the NPDEs and the *solid black lines* depict the 5th, 50th, and 95th percentiles of the standard normal distribution. *NPDEs* normalized prediction distribution errors




Fig. S7 Plots of urinary paracetamol-glucuronide NPDEs for evaluation of the final model. Density histogram of NPDEs (a) with overlaid *solid black curve* depicting the standard normal distribution. NPDEs versus **b** time since first dose, **c** population-predicted amount, **d** weight, and **e** urine flow rate. In **b–e**, the *dashed red lines* depict the 5th, 50th, and 95th percentiles of the NPDEs and the *solid black lines* depict the 5th, 50th, and 95th percentiles of the standard normal distribution. *NPDEs* normalized prediction distribution errors




Fig. S8 Plots of urinary paracetamol-sulfate NPDEs for evaluation of the final model. Density histogram of NPDEs (a) with overlaid *solid black curve* depicting the standard normal distribution. NPDEs versus **b** time since first dose, **c** population-predicted amount, **d** weight, and **e** urine flow rate. In **b–e**, the *dashed red lines* depict the 5th, 50th, and 95th percentiles of the NPDEs and the *solid black lines* depict the 5th, 50th, and 95th percentiles of the standard normal distribution. *NPDEs* normalized prediction distribution errors




Fig. S9 Plots of urinary oxidative pathway metabolite NPDEs for evaluation of the final model. Density histogram of NPDEs (a) with overlaid *solid black curve* depicting the standard normal distribution. NPDEs versus **b** time since first dose, **c** population-predicted amount, **d** weight, and **e** urine flow rate. In **b–e**, the *dashed red lines* depict the 5th, 50th, and 95th percentiles of the NPDEs and the *solid black lines* depict the 5th, 50th, and 95th percentiles of the standard normal distribution. *NPDEs* normalized prediction distribution errors

Appendix: NONMEM 7.2 code for the final model

\$PROBLEM Paracetamol and metabolites in plasma and urine, Final model

\$INPUT

C ; For IGNORE
ID ; Subject identification number
TIME ; Time (h)
AMT ; Dose (mg)
RATE ; Drug infusion rate (coded as -2)
DUR ; Drug infusion duration (h)
EVID ; Event identification
MDV ; Missing dependent variable
CMT ; Compartment (1=paracetamol in plasma, 2=paracetamol-glucuronide in plasma,
; 3=paracetamol-sulfate in plasma, 4=oxidative pathway metabolites in plasma,
; 5=paracetamol-glucuronide in urine, 6=paracetamol-sulfate in urine,
; 7=oxidative pathway metabolites in urine, 8=paracetamol in urine)
DV ; Dependent variable (Concentration, mg paracetamol equivalents/L)
L2 ; Level 2 data item for multivariate observations
UVOL ; Urine sample volume (mL)
UFLOW ; Average flow rate for each collected urine sample (mL/h)
WT ; Patient body weight (kg)
PNA ; Patient postnatal age (days)
SURG ; Primary indication (1=postoperative analgesia, 0=procedural analgesia)

\$DATA FinalData.CSV IGNORE=C

\$SUBROUTINE ADVAN6 TOL=6

\$MODEL

COMP=(CENTPRNT) ; Central compartment for parent drug (paracetamol) in plasma
COMP=(CENTGLUC) ; Central compartment for paracetamol-glucuronide in plasma
COMP=(CENTSULF) ; Central compartment for paracetamol-sulfate in plasma
COMP=(CENTOX) ; Central compartment for oxidative pathway metabolites in plasma
COMP=(UGLUC INITIALOFF) ; Urinary compartment for paracetamol-glucuronide
COMP=(USULF INITIALOFF) ; Urinary compartment for paracetamol-sulfate
COMP=(UOX INITIALOFF) ; Urinary compartment for oxidative pathway metabolites
COMP=(UPRNT INITIALOFF) ; Urinary compartment for parent drug (paracetamol)

\$PK

; Define values for centering covariates

MNWT = 2.3 ; Mean patient body weight (kg)
MNPNA = 7.5 ; Mean patient postnatal age (days)
MDUFLOW = 6.5 ; Median urine flow rate (mL/h)

; Define typical values with covariate effects

; Renal clearance for unchanged paracetamol
TVCL1= THETA(1) * (WT/MNWT)**THETA(12) * EXP(THETA(23) * (UFLOW-MDUFLOW)) * (1+THETA(29) * SURG)

; Volume of distribution for paracetamol
TVV1= THETA(2) * (WT/MNWT)**THETA(13)

; Formation clearance for paracetamol-glucuronide
TVCL2= THETA(3) * (WT/MNWT)**THETA(14) * (PNA/MNPNA)**THETA(27)

; Volume of distribution for paracetamol-glucuronide
TVV2= THETA(4) * (WT/MNWT)**THETA(15)

; Formation clearance for paracetamol-sulfate

```

TVCL3= THETA(5) * (WT/MNWT)**THETA(16)

; Volume of distribution for paracetamol-sulfate
TVV3= THETA(6) * (WT/MNWT)**THETA(17)

; Formation clearance for oxidative pathway metabolites
TVCL4= THETA(7) * (WT/MNWT)**THETA(18) * (PNA/MNPNA)**THETA(28)

; Volume of distribution for oxidative pathway metabolites
TVV4= THETA(8) * (WT/MNWT)**THETA(19)

; Renal clearance for paracetamol-glucuronide
TVCL5= THETA(9) * (WT/MNWT)**THETA(20) * EXP(THETA(24)*(UFLOW-MDUFLOW))

; Renal clearance for paracetamol-sulfate
TVCL6= THETA(10) * (WT/MNWT)**THETA(21) * EXP(THETA(25)*(UFLOW-MDUFLOW))

; Renal clearance for oxidative pathway metabolites
TVCL7= THETA(11) * (WT/MNWT)**THETA(22) * EXP(THETA(26)*(UFLOW-MDUFLOW))

; Define individual values
CL1 = TVCL1 * EXP(ETA(1)) ; Renal CL for unchanged paracetamol
V1 = TVV1 * EXP(ETA(2)) ; Vd for paracetamol
CL2 = TVCL2 * EXP(ETA(3)) ; Formation CL for paracetamol-glucuronide
V2 = TVV2 * EXP(ETA(4)) ; Vd for paracetamol-glucuronide
CL3 = TVCL3 * EXP(ETA(5)) ; Formation CL for paracetamol-sulfate
V3 = TVV3 * EXP(ETA(6)) ; Vd for paracetamol-sulfate
CL4 = TVCL4 * EXP(ETA(7)) ; Formation CL for oxidative pathway metabolites
V4 = TVV4 * EXP(ETA(8)) ; Vd for oxidative pathway metabolites
CL5 = TVCL5 * EXP(ETA(9)) ; Renal CL for paracetamol-glucuronide

; Define individual value for renal CL of paracetamol-sulfate
; and fix correlation in BSV for renal clearances of
; paracetamol-sulfate and paracetamol-glucuronide to 1
CL6 = TVCL6 * EXP(THETA(30)*ETA(9))

; Define individual value for renal CL of oxidative pathway metabolites
; and fix correlation in BSV for renal clearances of
; oxidative pathway metabolites and paracetamol-glucuronide to 1
CL7 = TVCL7 * EXP(THETA(31)*ETA(9))

D1 = DUR ; Define data item for drug infusion duration (h)

S1 = V1
S2 = V2
S3 = V3
S4 = V4
S5 = UVOL/1000
S6 = UVOL/1000
S7 = UVOL/1000
S8 = UVOL/1000

K18 = CL1 / V1
K12 = CL2 / V1
K13 = CL3 / V1
K14 = CL4 / V1
K25 = CL5 / V2
K36 = CL6 / V3
K47 = CL7 / V4

$DES

DADT(1) = -A(1) * (K18 + K12 + K13 + K14)

```

```
DADT(2) = A(1) * K12 - A(2) * K25
DADT(3) = A(1) * K13 - A(3) * K36
DADT(4) = A(1) * K14 - A(4) * K47
DADT(5) = A(2) * K25
DADT(6) = A(3) * K36
DADT(7) = A(4) * K47
DADT(8) = A(1) * K18
```

```
$ERROR
```

```
IPRED = F
```

```
; Define indicator variables so RUV can be estimated separately for each CMT
```

```
CT1 = 0
CT2 = 0
CT3 = 0
CT4 = 0
CT5 = 0
CT6 = 0
CT7 = 0
CT8 = 0
```

```
IF(CMT.EQ.1) CT1=1
IF(CMT.EQ.2) CT2=1
IF(CMT.EQ.3) CT3=1
IF(CMT.EQ.4) CT4=1
IF(CMT.EQ.5) CT5=1
IF(CMT.EQ.6) CT6=1
IF(CMT.EQ.7) CT7=1
IF(CMT.EQ.8) CT8=1
```

```
; Define proportional RUV for all compartments
```

```
PROPP = CT1*ERR(1) + CT2*ERR(2) + CT3*ERR(3) + CT4*ERR(4) ; plasma compartments
PROPU = CT5*ERR(5) + CT6*ERR(6) + CT7*ERR(7) + CT8*ERR(8) ; urinary compartments
PROPERR = 1 + PROPP + PROPU
```

```
; Define additive RUV for paracetamol-sulfate in plasma and urine
```

```
; and for paracetamol in urine
```

```
; And fix correlation in RUV for paracetamol-sulfate and paracetamol in urine to 1
```

```
ADDERR = CT3*ERR(9) + CT6*ERR(10) + CT8*(THETA(32)*ERR(10))
```

```
Y = IPRED * (PROPERR) + ADDERR
```

```
$ESTIMATION METHOD=1 INTERACTION PRINT=5 MAX=99999 NSIG=2 SIGL=6 NOABORT
```

```
; Define initial estimates
```

```
$THETA
```

```
( 0, 0.02, 1) ; CL1
( 0, 3, 15) ; V1
( 0, 0.05, 1) ; CL2
( 0, 1, 15) ; V2
( 0, 0.2, 2) ; CL3
( 0, 1, 15) ; V3
( 0, 0.06, 1) ; CL4
( 0, 3, 15) ; V4
( 0, 0.1, 2) ; CL5
( 0, 0.1, 2) ; CL6
( 0, 0.2, 2) ; CL7
(-1, 1, 5) ; WT on CL1
(-1, 1, 5) ; WT on V1
(-1, 1, 5) ; WT on CL2
```

```

(-1, 1, 5) ; WT on V2
(-1, 1, 5) ; WT on CL3
(-1, 1, 5) ; WT on V3
(-1, 1, 5) ; WT on CL4
(-1, 1, 5) ; WT on V4
(-1, 1, 5) ; WT on CL5
(-1, 1, 5) ; WT on CL6
(-1, 1, 5) ; WT on CL7
(-1, 0.02, 1) ; UFLOW on CL1
(-1, 0.02, 1) ; UFLOW on CL5
(-1, 0.02, 1) ; UFLOW on CL6
(-1, 0.02, 1) ; UFLOW on CL7
(-1, 1, 5) ; PNA on CL2
(-1, 1, 5) ; PNA on CL4
(-2, -0.5, 2) ; SURG on CL1
(1.5) ; Scale parameter between variances of renal CL for paracetamol-glucuronide
; and renal CL for paracetamol-sulfate (see Equations 5 and 6)
(1.2) ; Scale parameter between variances of renal CL for paracetamol-glucuronide
; and renal CL for oxidative pathway metabolites (see Equations 5 and 6)
(0.01) ; Scale parameter between variances of additive RUV for paracetamol in urine
; and additive RUV for paracetamol-sulfate in urine (see Equations 5 and 6)

```

\$OMEGA BLOCK(9)

```

0.3 ; CL1
0.01 0.3 ; V1
0.01 0.01 0.3 ; CL2
0.01 0.01 0.01 0.3 ; V2
0.01 0.01 0.01 0.01 0.3 ; CL3
0.01 0.01 0.01 0.01 0.01 0.3 ; V3
0.01 0.01 0.01 0.01 0.01 0.01 0.3 ; CL4
0.01 0.01 0.01 0.01 0.01 0.01 0.01 0.3 ; V4
0.01 0.01 0.01 0.01 0.01 0.01 0.01 0.01 0.3 ; CL5

```

```

; Proportional RUV for plasma compartments
; with covariance estimated for multivariate observations

```

\$\$SIGMA BLOCK(4)

```

0.1 ; plasma paracetamol
0.01 0.1 ; plasma paracetamol-glucuronide
0.01 0.01 0.1 ; plasma paracetamol-sulfate
0.01 0.01 0.01 0.1 ; plasma oxidative pathway metabolites

```

```

; Proportional RUV for urine compartments
; with covariance estimated for multivariate observations

```

\$\$SIGMA BLOCK(4)

```

0.4 ; urinary paracetamol-glucuronide
0.1 0.4 ; urinary paracetamol-sulfate
0.1 0.1 0.4 ; urinary oxidative pathway metabolites
0.1 0.1 0.1 0.4 ; urinary paracetamol

```

```

; Additive RUV

```

\$\$SIGMA

```

1.2 ; plasma paracetamol-sulfate
200 ; urinary paracetamol-sulfate

```